# NAG Library Routine Document

# S30NBF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

S30NBF computes the European option price given by Heston's stochastic volatility model together with its sensitivities (Greeks).

## 2 Specification

```
SUBROUTINE S30NBF (CALPUT, M, N, X, S, T, SIGMAV, KAPPA, CORR, VAR0, ETA,      &
                   GRISK, R, Q, P, LDP, DELTA, GAMMA, VEGA, THETA, RHO,        &
                   VANNA, CHARM, SPEED, ZOMMA, VOMMA, IFAIL)

INTEGER            M, N, LDP, IFAIL
REAL (KIND=nag_wp) X(M), S, T(N), SIGMAV, KAPPA, CORR, VAR0, ETA, GRISK,       &
                   R, Q, P(LDP,N), DELTA(LDP,N), GAMMA(LDP,N),                 &
                   VEGA(LDP,N), THETA(LDP,N), RHO(LDP,N), VANNA(LDP,N),        &
                   CHARM(LDP,N), SPEED(LDP,N), ZOMMA(LDP,N), VOMMA(LDP,N)
CHARACTER(1)       CALPUT
```

## 3 Description

S30NBF computes the price and sensitivities of a European option using Heston's stochastic volatility model. The return on the asset price, $S$, is

$$\frac{dS}{S} = (r-q)dt + \sqrt{v_t}dW_t^{(1)}$$

and the instantaneous variance, $v_t$, is defined by a mean-reverting square root stochastic process,

$$dv_t = \kappa(\eta - v_t)dt + \sigma_v\sqrt{v_t}dW_t^{(2)},$$

where $r$ is the risk free annual interest rate; $q$ is the annual dividend rate; $v_t$ is the variance of the asset price; $\sigma_v$ is the volatility of the volatility, $\sqrt{v_t}$; $\kappa$ is the mean reversion rate; $\eta$ is the long term variance. $dW_t^{(i)}$, for $i = 1, 2$, denotes two correlated standard Brownian motions with

$$\mathbb{C}\mathrm{ov}\left[dW_t^{(1)}, dW_t^{(2)}\right] = \rho dt.$$

The option price is computed by evaluating the integral transform given by Lewis (2000) using the form of the characteristic function discussed by Albrecher *et al.* (2007), see also Kilin (2006).

$$P_{\text{call}} = Se^{-qT} - Xe^{-rT}\frac{1}{\pi}\mathrm{Re}\left[\int_{0+i/2}^{\infty+i/2} e^{-ik\bar{X}}\frac{\hat{H}(k,v,T)}{k^2-ik}\mathrm{d}k\right], \tag{1}$$

where $\bar{X} = \ln(S/X) + (r-q)T$ and

$$\hat{H}(k,v,T) = \exp\left(\frac{2\kappa\eta}{\sigma_v^2}\left[tg \; - \ln\left(\frac{1-he^{-\xi t}}{1-h}\right)\right] + v_t g\left[\frac{1-e^{-\xi t}}{1-he^{-\xi t}}\right]\right),$$

$$g = \frac{1}{2}(b-\xi), \qquad h = \frac{b-\xi}{b+\xi}, \qquad t = \sigma_v^2 T/2,$$

$$\xi = \left[b^2 + 4\frac{k^2-ik}{\sigma_v^2}\right]^{\frac{1}{2}},$$

$$b = \frac{2}{\sigma_v^2}\left[(1-\gamma+ik)\rho\sigma_v + \sqrt{\kappa^2-\gamma(1-\gamma)\sigma_v^2}\right]$$

with $t = \sigma_v^2 T/2$. Here $\gamma$ is the risk aversion parameter of the representative agent with $0 \leq \gamma \leq 1$ and $\gamma(1-\gamma)\sigma_v^2 \leq \kappa^2$. The value $\gamma = 1$ corresponds to $\lambda = 0$, where $\lambda$ is the market price of risk in Heston (1993) (see Lewis (2000) and Rouah and Vainberg (2007)).

The price of a put option is obtained by put-call parity.

$$P_{\text{put}} = P_{\text{call}} + Xe^{-rT} - Se^{-qT}.$$

Writing the expression for the price of a call option as

$$P_{\text{call}} = Se^{-qT} - Xe^{-rT}\frac{1}{\pi}\text{Re}\left[\int_{0+i/2}^{\infty+i/2} I(k,r,S,T,v)\text{d}k\right]$$

then the sensitivities or Greeks can be obtained in the following manner,

Delta

$$\frac{\partial P_{\text{call}}}{\partial S} = e^{-qT} + \frac{Xe^{-rT}}{S}\frac{1}{\pi}\text{Re}\left[\int_{0+i/2}^{\infty+i/2}(ik)I(k,r,S,T,v)\text{d}k\right],$$

Vega

$$\frac{\partial P}{\partial v} = -Xe^{-rT}\frac{1}{\pi}\text{Re}\left[\int_{0-i/2}^{0+i/2} f_2 I(k,r,j,S,T,v)\text{d}k\right], \qquad \text{where } f_2 = g\left[\frac{1-e^{-\xi t}}{1-he^{-\xi t}}\right],$$

Rho

$$\frac{\partial P_{\text{call}}}{\partial r} = TXe^{-rT}\frac{1}{\pi}\text{Re}\left[\int_{0+i/2}^{\infty+i/2}(1+ik)I(k,r,S,T,v)\text{d}k\right].$$

## 4 References

Albrecher H, Mayer P, Schoutens W and Tistaert J (2007) The little Heston trap *Wilmott Magazine* **January 2007** 83–92

Heston S (1993) A closed-form solution for options with stochastic volatility with applications to bond and currency options **6** 347–343 Review of Financial Studies

Kilin F (2006) Accelerating the calibration of stochastic volatility models *MPRA Paper No. 2975* http://mpra.ub.uni-muenchen.de/2975/

Lewis A L (2000) Option valuation under stochastic volatility *Finance Press, USA*

Rouah F D and Vainberg G (2007) *Option Pricing Models and Volatility using Excel-VBA* John Wiley and Sons, Inc

## 5 Parameters

1: CALPUT – CHARACTER(1) *Input*

*On entry*: determines whether the option is a call or a put.

CALPUT = 'C'
A call. The holder has a right to buy.

CALPUT = 'P'
A put. The holder has a right to sell.

*Constraint*: CALPUT = 'C' or 'P'.

2: M – INTEGER *Input*

*On entry*: the number of strike prices to be used.

*Constraint*: M $\geq 1$.

3: N – INTEGER *Input*

*On entry*: the number of times to expiry to be used.

*Constraint*: N $\geq 1$.

4: X(M) – REAL (KIND=nag_wp) array *Input*

*On entry*: X($i$) must contain $X_i$, the $i$th strike price, for $i = 1, 2, \ldots,$ M.

*Constraint*: X($i$) $\geq z$ and X($i$) $\leq 1/z$, where $z =$ X02AMF(), the safe range parameter, for $i = 1, 2, \ldots,$ M.

5: S – REAL (KIND=nag_wp) *Input*

*On entry*: $S$, the price of the underlying asset.

*Constraint*: S $\geq z$ and S $\leq 1.0/z$, where $z =$ X02AMF(), the safe range parameter.

6: T(N) – REAL (KIND=nag_wp) array *Input*

*On entry*: T($i$) must contain $T_i$, the $i$th time, in years, to expiry, for $i = 1, 2, \ldots,$ N.

*Constraint*: T($i$) $\geq z$, where $z =$ X02AMF(), the safe range parameter, for $i = 1, 2, \ldots,$ N.

7: SIGMAV – REAL (KIND=nag_wp) *Input*

*On entry*: the volatility, $\sigma_v$, of the volatility process, $\sqrt{v_t}$. Note that a rate of 20% should be entered as 0.2.

*Constraint*: SIGMAV $> 0.0$.

8: KAPPA – REAL (KIND=nag_wp) *Input*

*On entry*: $\kappa$, the long term mean reversion rate of the volatility.

*Constraint*: KAPPA $> 0.0$.

9: CORR – REAL (KIND=nag_wp) *Input*

*On entry*: the correlation between the two standard Brownian motions for the asset price and the volatility.

*Constraint*: $-1.0 \leq$ CORR $\leq 1.0$.

10: VAR0 – REAL (KIND=nag_wp) *Input*

*On entry*: the initial value of the variance, $v_t$, of the asset price.

*Constraint*: VAR0 $\geq 0.0$.

11: ETA – REAL (KIND=nag_wp) *Input*

*On entry*: $\eta$, the long term mean of the variance of the asset price.

*Constraint*: ETA $> 0.0$.

12: GRISK – REAL (KIND=nag_wp) *Input*

*On entry*: the risk aversion parameter, $\gamma$, of the representative agent.

*Constraint*: $0.0 \leq \text{GRISK} \leq 1.0$ and $\text{GRISK} \times (1 - \text{GRISK}) \times \text{SIGMAV} \times \text{SIGMAV} \leq \text{KAPPA} \times \text{KAPPA}$.

13: R – REAL (KIND=nag_wp) *Input*

*On entry*: $r$, the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.

*Constraint*: $\text{R} \geq 0.0$.

14: Q – REAL (KIND=nag_wp) *Input*

*On entry*: $q$, the annual continuous yield rate. Note that a rate of 8% should be entered as 0.08.

*Constraint*: $\text{Q} \geq 0.0$.

15: P(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\text{M} \times \text{N}$ part of the array P contains the computed option prices.

16: LDP – INTEGER *Input*

*On entry*: the first dimension of the arrays P, DELTA, GAMMA, VEGA, THETA, RHO, VANNA, CHARM, SPEED, ZOMMA and VOMMA as declared in the (sub)program from which S30NBF is called.

*Constraint*: $\text{LDP} \geq \text{M}$.

17: DELTA(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\text{M} \times \text{N}$ part of the array DELTA contains the sensitivity, $\frac{\partial P}{\partial S}$, of the option price to change in the price of the underlying asset.

18: GAMMA(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\text{M} \times \text{N}$ part of the array GAMMA contains the sensitivity, $\frac{\partial^2 P}{\partial S^2}$, of DELTA to change in the price of the underlying asset.

19: VEGA(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\text{M} \times \text{N}$ part of the array VEGA contains the sensitivity, $\frac{\partial P}{\partial \sigma}$, of the option price to change in the volatility of the underlying asset.

20: THETA(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\text{M} \times \text{N}$ part of the array THETA contains the sensitivity, $-\frac{\partial P}{\partial T}$, of the option price to change in the time to expiry of the option.

21: RHO(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\text{M} \times \text{N}$ part of the array RHO contains the sensitivity, $\frac{\partial P}{\partial r}$, of the option price to change in the annual risk-free interest rate.

22: VANNA(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\text{M} \times \text{N}$ part of the array VANNA contains the sensitivity, $\frac{\partial^2 P}{\partial S \partial \sigma}$, of VEGA to change in the price of the underlying asset or, equivalently, the sensitivity of DELTA to change in the volatility of the asset price.

23: CHARM(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\mathrm{M} \times \mathrm{N}$ part of the array CHARM contains the sensitivity, $-\frac{\partial^2 P}{\partial S \partial T}$, of DELTA to change in the time to expiry of the option.

24: SPEED(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\mathrm{M} \times \mathrm{N}$ part of the array SPEED contains the sensitivity, $\frac{\partial^3 P}{\partial S^3}$, of GAMMA to change in the price of the underlying asset.

25: ZOMMA(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\mathrm{M} \times \mathrm{N}$ part of the array ZOMMA contains the sensitivity, $\frac{\partial^3 P}{\partial S^2 \partial \sigma}$, of GAMMA to change in the volatility of the underlying asset.

26: VOMMA(LDP,N) – REAL (KIND=nag_wp) array *Output*

*On exit*: the leading $\mathrm{M} \times \mathrm{N}$ part of the array VOMMA contains the sensitivity, $\frac{\partial^2 P}{\partial \sigma^2}$, of VEGA to change in the volatility of the underlying asset.

27: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to 0, $-1$ or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value $-1$ or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit*: $\mathrm{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry $\mathrm{IFAIL} = 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\mathrm{IFAIL} = 1$

On entry, CALPUT $\neq$ 'C' or 'P'.

$\mathrm{IFAIL} = 2$

On entry, $\mathrm{M} \leq 0$.

$\mathrm{IFAIL} = 3$

On entry, $\mathrm{N} \leq 0$.

$\mathrm{IFAIL} = 4$

On entry, $\mathrm{X}(i) < z$ or $\mathrm{X}(i) > 1/z$, where $z = $ X02AMF(), the safe range parameter.

$\mathrm{IFAIL} = 5$

On entry, $\mathrm{S} < z$ or $\mathrm{S} > 1.0/z$, where $z = $ X02AMF(), the safe range parameter.

IFAIL $= 6$

On entry, $\mathrm{T}(i) < z$, where $z = \mathrm{X02AMF}()$, the safe range parameter.

IFAIL $= 7$

On entry, SIGMAV $\leq 0.0$.

IFAIL $= 8$

On entry, KAPPA $\leq 0.0$.

IFAIL $= 9$

On entry, $|\mathrm{CORR}| > 1.0$.

IFAIL $= 10$

On entry, VAR0 $< 0.0$.

IFAIL $= 11$

On entry, ETA $\leq 0.0$.

IFAIL $= 12$

On entry, GRISK $< 0.0$ or GRISK $> 1.0$,
or GRISK $\times$ (1.0 $-$ GRISK) $\times$ SIGMAV $\times$ SIGMAV $>$ KAPPA $\times$ KAPPA.

IFAIL $= 13$

On entry, R $< 0.0$.

IFAIL $= 14$

On entry, Q $< 0.0$.

IFAIL $= 16$

On entry, LDP $<$ M.

IFAIL $= 17$

Quadrature has not converged to the required accuracy. However the result should be a reasonable approximation.

IFAIL $= 18$

Quadrature has not converged to the required accuracy. The values returned cannot be relied upon.

## 7 Accuracy

The accuracy of the output is determined by the accuracy of the numerical quadrature used to evaluate the integral in (1). An adaptive method is used which evaluates the integral to within a tolerance of $\max\left(10^{-8}, 10^{-10} \times |I|\right)$, where $|I|$ is the absolute value of the integral.

## 8 Further Comments

None.

## 9 Example

This example computes the price and sensitivities of a European call using Heston's stochastic volatility model. The time to expiry is 1 year, the stock price is 100 and the strike price is 100. The risk-free interest rate is 2.5% per year, the volatility of the variance, $\sigma_v$, is 57.51% per year, the mean reversion parameter, $\kappa$, is 1.5768, the long term mean of the variance, $\eta$, is 0.0398 and the correlation between the volatility process and the stock price process, $\rho$, is $-0.5711$. The risk aversion parameter, $\gamma$, is 1.0 and the initial value of the variance, VAR0, is 0.0175.

## 9.1 Program Text

```
    Program s30nbfe

!     S30NBF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s30nbf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: corr, eta, grisk, kappa, q, r, s,    &
                                          sigmav, var0
      Integer                          :: i, ifail, j, ldp, m, n
      Character (1)                    :: calput
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: charm(:,:), delta(:,:), gamma(:,:),  &
                                          p(:,:), rho(:,:), speed(:,:), t(:),  &
                                          theta(:,:), vanna(:,:), vega(:,:),   &
                                          vomma(:,:), x(:), zomma(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'S30NBF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) calput
      Read (nin,*) s, r, q
      Read (nin,*) kappa, eta, var0, sigmav, corr, grisk
      Read (nin,*) m, n

      ldp = m
      Allocate (charm(ldp,n),delta(ldp,n),gamma(ldp,n),p(ldp,n),rho(ldp,n), &
        speed(ldp,n),t(n),theta(ldp,n),vanna(ldp,n),vega(ldp,n),vomma(ldp,n), &
        x(m),zomma(ldp,n))

      Read (nin,*)(x(i),i=1,m)
      Read (nin,*)(t(i),i=1,n)

      ifail = 0
      Call s30nbf(calput,m,n,x,s,t,sigmav,kappa,corr,var0,eta,grisk,r,q,p,ldp, &
        delta,gamma,vega,theta,rho,vanna,charm,speed,zomma,vomma,ifail)

      Write (nout,*)
      Write (nout,*) 'Heston''s Stochastic volatility Model'

      Select Case (calput)
      Case ('C','c')
        Write (nout,*) 'European Call :'
      Case ('P','p')
        Write (nout,*) 'European Put :'
      End Select

      Write (nout,99997) '  Spot                     = ', s
      Write (nout,99997) '  Volatility of vol        = ', sigmav
      Write (nout,99997) '  Mean reversion           = ', kappa
      Write (nout,99997) '  Correlation              = ', corr
      Write (nout,99997) '  Variance                 = ', var0
      Write (nout,99997) '  Mean of variance         = ', eta
      Write (nout,99997) '  Risk aversion            = ', grisk
```

```
      Write (nout,99997) '  Rate                      = ', r
      Write (nout,99997) '  Dividend                  = ', q

      Write (nout,*)

      Do j = 1, n
        Write (nout,*)
        Write (nout,99999) t(j)
        Write (nout,*) &
          '  Strike       Price      Delta      Gamma      Vega       ' // &
          'Theta      Rho'

        Do i = 1, m
          Write (nout,99998) x(i), p(i,j), delta(i,j), gamma(i,j), vega(i,j), &
            theta(i,j), rho(i,j)
        End Do

        Write (nout,*) &
          '  Strike       Price      Vanna      Charm      Speed      ' // &
          'Zomma      Vomma'

        Do i = 1, m
          Write (nout,99998) x(i), p(i,j), vanna(i,j), charm(i,j), speed(i,j), &
            zomma(i,j), vomma(i,j)
        End Do

      End Do


99999 Format (1X,'Time to Expiry : ',1X,F8.4)
99998 Format (1X,7(F10.4,1X))
99997 Format (A,1X,F10.4)
    End Program s30nbfe
```

## 9.2 Program Data

```
S30NBF Example Program Data
 'C'                          : Call = 'C', Put = 'P'
 100.0 0.025 0.0               : S, R, Q
 1.5768 0.0398 0.0175 0.5751 -0.5711 1.0 : KAPPA, ETA, VAR0, SIGMAV, CORR, GRISK
 1 1                          : M, N
 100.0                        : X(I), I = 1,2,...N
 1.0                          : T(I), I = 1,2,...M
```

## 9.3 Program Results

```
 S30NBF Example Program Results

 Heston's Stochastic volatility Model
 European Call :
  Spot                   =    100.0000
  Volatility of vol      =      0.5751
  Mean reversion         =      1.5768
  Correlation            =     -0.5711
  Variance               =      0.0175
  Mean of variance       =      0.0398
  Risk aversion          =      1.0000
  Rate                   =      0.0250
  Dividend               =      0.0000


 Time to Expiry :    1.0000
   Strike       Price      Delta      Gamma       Vega       Theta       Rho
   100.0000     7.2743     0.6945     0.0251    52.5461    -4.9969    62.1735
   Strike       Price      Vanna      Charm      Speed      Zomma      Vomma
   100.0000     7.2743    -0.5643    -0.0321    -0.0023    -0.1976  -321.0780
```